

Graphics with Processing



2019-12 レンダリング技術

<http://vilab.org>


塩澤秀和

12.1* 影付け

影の種類 (p.158)

□ 本影と半影

- 点光源や平行光ではくっきりした影(本影)だけができる
- 光源に広がりがあると,半影を含むソフトシャドウができる




- 光源が複数ある場合,それぞれの光(影)を重ね合せばよい
- リアルタイムな影生成では基本的に本影部分を扱う

主な影付け方式

□ 影の投影テクスチャマッピング


- いったん視点を光源に置き,物体のシルエットを描画すると,光源から見たその物体の影になる
- 視点は戻して,影の画像を光源の位置から物体の下の地面などに投影テクスチャマッピングする



12.2* 影付け(続き)


□ シャドウボリューム法

- 物体が光をさえぎってできる影の空間(シャドウボリューム)を囲む“影ポリゴン”を算出する
- 視点から見て表を向いている影ポリゴンの像から,裏を向いている影ポリゴンの像を引くと,視点から見た影が描画できる
- 「ステンシル(型抜き)バッファ」を用いると,高速に実現できる



□ シャドウマップ法(p.159)

- (Zバッファを用いた2段階法)
- 視点を光源に置き,Zバッファだけを描画すると,光の到達距離Sの分布図(シャドウマップ)ができる
- 視点を戻し,描画対象の3D座標から光源までの距離Lを計算し,シャドウマップ上の対応点の値Sと比較すると,描画対象まで光が届いているか判定できる



12.3 もっと単純な影の例

```
float x = 0, y = -200, z = -200;

void draw() {
  background(50, 50, 100);
  perspective();
  camera(-150, -500, 500,
         0, 0, 0, 0, 1, 0);
  directionalLight(200, 200, 200,
                  0, 1, 0);
  ambientLight(128, 128, 128);

  z++;

  fill(0, 150, 0);
  beginShape(QUADS);
  float w = 500;
  vertex(-w, 0, -w); vertex(-w, 0, w);
  vertex(w, 0, w); vertex(w, 0, -w);
  endShape();

  // 本体の表示
  pushMatrix();
  fill(255);
  drawObjects();
  popMatrix();

  // 縦方向に潰して真上からの影を作成
  pushMatrix();
  fill(0, 180); // 黒色(半透明)
  translate(0, -1, 0); // 地面の少し上
  scale(1, 0.1, 1); // y方向に潰す
  drawObjects();
  popMatrix();
}

void drawObjects() {
  translate(x, y, z);
  rotateX(PI/3); rotateY(PI/6);
  box(100);
}
```

12.4* 高品質レンダリング

目的別レンダリング

- リアルタイムレンダリング
 - 3Dゲーム ← ユーザが操作
 - 理想は60fps, 最低限10fps
- 高品質レンダリング
 - 静止画, 映画 ← 事前に“撮影”
 - やわらかい陰影やガラスの表現
 - ⇒ レイトレーシング法+大域照明

大域照明モデル (p.183)

(Global Illumination: GI)

- 間接光まで含む照明計算
 - 単純な環境光モデルではなく, 間接光をより精密に計算する
 - 特に室内の陰影がより自然
 - ラジオシティ, フォトンマッピング

フリーソフトによるレンダリングの例

- POV-Ray
 - <http://www.povray.org>
 - Hall of Fame
- Blender+Yafray
 - <http://www.blender.org>
 - Feature & Gallery
 - <http://www.yafaray.org>
 - Gallery
- Sunflow
 - <http://sunflow.sourceforge.net>
 - Gallerly (開発終了?)
- Art of Illusion
 - <http://www.artofillusion.org>
 - Art Gallery

12.5* レイトレーシング (p.135)


レイトレーシング(光線追跡)法

□ 概要


- 画面の各画素に届く光線(レイ)を視点から逆方向に追跡する
- 視点から、各画素に対応するレイ(半直線)を“飛ばす”
- レイが物体と交差(衝突)したら、材質と照明から画素の色を求める
- 影を描画する場合、衝突点から光源にレイを飛ばして判定する
- 鏡面反射, 透過・屈折を扱う場合、レイを分岐して再帰的に追跡する

□ 特徴

- 隠面消去や影付けが容易
- 透明, レンズ, 映り込み等も再現
- 映像作品(映画等)では一般的
- まだリアルタイム処理には不向き



光線の分岐



12.6 レイを飛ばす処理の基本

```
// レイトレーシングの基本(レイキャスティング)
// によるレイと球の交差判定の例
import static processing.core.PVector.*;

void setup() { size(600, 600); noLoop(); }


// 視点座標系における球の中心と半径
PVector center = new PVector(0, 0, -10);
float r = 1.0;
// 照明(方向光)の方向ベクトル
PVector light =
  new PVector(1, 1, -3).normalize();

void draw() {
  // 全ピクセルに対し、レイを飛ばして画面描画
  loadPixels();
  for (int x = 0; x < width; x++)
    for (int y = 0; y < height; y++)
      pixels[y * width + x] = raycast(x, y);
  updatePixels();
}
```

```
color raycast(int x, int y) {
  // 視点座標系で視点(原点)の前にスクリーンを想定
  float scrX = (x * 2.0 - width) / width;
  float scrY = (y * 2.0 - height) / height;
  float scrZ = -2.0;
  // 視点から仮想スクリーンの点の方向にレイを飛ばす
  PVector ray = new PVector(scrX, scrY, scrZ);
  ray.normalize();
  // レイの延長線上で球の中心に最も近づく点を求める
  PVector nearest = mult(ray, center.dot(ray));
  // その点が球の内側なら交差あり(効率優先の計算式)
  float d2 = r * r - sub(nearest, center).magSq();
  if (d2 > 0) {
    // 球面上の交点とそこでの法線ベクトルを求める
    PVector p = sub(nearest, mult(ray, sqrt(d2)));
    PVector n = sub(p, center).normalize();
    // ランバート反射によるシェーディング計算
    float f = -n.dot(light);
    if (f > 0) return color(f * 255);
  }
  return color(0);
}
```

12.7 レイを飛ばす処理の基本 (補足)

12.6の図解



PVectorクラス

コンストラクタ	<code>v = PVector(x, y, z)</code>
加算	<code>v.add(u)</code> <code>w = PVector.add(v, u)</code>
減算	<code>v.sub(u)</code> <code>w = PVector.sub(v, u)</code>
スカラー倍	<code>v.mult(s)</code> <code>w = PVector.mult(v, s)</code>
内積	<code>s = v.dot(u)</code>
大きさ(ノルム)	<code>s = v.mag()</code>
ノルムの2乗	<code>s = v.magSq()</code>
正規化	<code>v.normalize()</code>

□ Java: 静的インポート

- `import static クラス名.*;`
- 静的フィールド/メソッド利用時に「クラス名.」が省略可能になる

12.8* フォトンマッピング (p.187)

フォトン(Photon)マッピング

□ 概要


- 光源から出る大量の光子を考え、その軌跡をシミュレーションする
- すると、シーン全体の光の分布(間接光)が概算できる
- この間接光を環境光の代わりに使って、レイトレーシングを行う

□ 特徴

- レンズなどの集光現象(コースティック)が表現できる
- 逆方向のレイトレーシングといえ、レイトレーシング法と相性がよい
- 着想は簡単だが、アルゴリズムは複雑で膨大な時間がかかる



Wikipedia



計算された光子の分布

12.9* ラジオシティ法 (p.184)

ラジオシティ(Radiosity)法

概要

- ポリゴンをパッチ (断片ポリゴン) に分割する
- 2つのパッチの位置と向きの関係から, 光の相互伝達率 (フォームファクタ) を計算する
- 全パッチ間での光エネルギーの放射発散の平衡状態を求める

ラジオシティ方程式 (p.158)

$$B_i = E_i + \rho_i \sum_{j=1}^n F_{ij} B_j$$

n シーン全体のパッチ数


B_i パッチ*i*の光の放射量 (ラジオシティ)

E_i パッチ*i*の発光量


ρ_i パッチ*i*の反射率

F_{ij} フォームファクタ ($F_{ij} = F_{ji}$)

- 巨大な「連立一次方程式」になる
⇒ コンピュータによる数値計算



柔らかい影や壁の色の影響が表現されている



12.10 その他のレンダリング技術

ぼかし(ボケ)系

- アンチエイリアシング(p.255)
 - ドットのギザギザが目立たないように,輪郭を中間色でぼかす
- フォグ(霧)
 - 水蒸気やチリなどによる空気の「濁り」を再現する
 - 遠くにあるものがかすんでいき,色が落ちていく効果を与える
- 被写界深度(DOF)(p.301)
 - レンズの効果を実現し,ピントが合っていないところをぼかす
- モーションブラー
 - 速く動くものに見える残像をわざと表示する
 - 軌跡の画像を重ね合わせる

イメージベースレンダリング

- 画像をCGに利用(p.171)
 - CGと画像処理技術との融合
 - テクスチャマッピングの応用(撮影地点から画像を投影など)
 - イメージベースライティング(IBL):画像を光源として利用
 - 環境マッピング:周辺の景色の映り込みを表現
 - イメージベースモデリング:写真から3Dモデルを自動生成
- 実写とCGの融合
 - 実写にCG映像を合成(AR),または,CGに実写映像を合成
 - 自由視点画像:限られた台数で撮影したカメラ映像から,自由な視点からの映像を合成

12.11* 非写実的レンダリング (p.309)

ノンフォトリアリスティック(非写実的) レンダリング(NPR)

□ 概要

- 現実の再現を目的としないCG
- 例) 油絵風, 手書きタッチの再現, 製図風, 2次元アニメ, 芸術作品

□ 背景

- 写実的(フォトリアリスティック)なCG技術はかなり完成
- 漫画・アニメーションでの利用
- 芸術などへのCG利用の広がり


□ Blender Freestyle

- フリーの3DCGソフトウェア
Blenderに付属のNPR機能

[http://www.blender.org/manual/
render/freestyle/introduction.html](http://www.blender.org/manual/render/freestyle/introduction.html)



Wikipedia



Wikipedia

12.12 演習課題

Processingでレイトレーシング

□ joons-renderer

- Sunflowを利用するライブラリ
- github.com/joonhyublee/joons-renderer/
- コンパイル済み → vilab.org/cg2019/joons102.zip
- ZIPを展開し, jonesrenderer フォルダをProcessingフォルダの中のlibrariesの中にコピー

□ 自由課題

- 12.12のプログラムを改造し, 適当な図形を表示させてみよ
- または, 12.6を改造し, 複数の球を表示させてみよ
- 今回の課題は提出しなくてよい

```

import joons.JoonsRenderer;
JoonsRenderer jr;

void setup() {
  size(800, 600, P3D);
  jr = new JoonsRenderer(this);
}

void draw() {
  jr.beginRecord();

  camera(0, 0, 120, 0, 0, -1, 0, 1, 0);
  perspective(PI/4, 4.0/3.0, 10, 1000);

  jr.background("cornell_box", 100, 100, 100);
  jr.background("gi_instant");

  jr.fill("diffuse", 255, 255, 255);
  translate(0, 10, -10);
  rotateY(-PI/8); rotateX(-PI/8);
  box(20);

  jr.endRecord();
  jr.displayRendered(true);
}

void keyPressed() {
  if (key == 'r' || key == 'R') jr.render();
}

```

レンダリング
結果を保存

色

図形
描画

Rキーでレン
ダリング開始